

1 The File garden_party.pl

```
:- use_module(library(clpfd)).
```

garden_party(Pairs, Dishes, Beverages)	garden_party.pl
--	-----------------

Four chefs, Aisha, Emma, Mei, and Winona, are attending a garden party. They each prepare a different dish and bring a different beverage.

```
garden_party(Pairs, Dishes, Beverages) :-  
    Table = [Chefs, Dishes, Beverages],  
    ChefNames = [aisha, emma, mei, winona],  
    Chefs = [Aisha, Emma, Mei, Winona],  
    Dishes = [PadThai, Frybread, Tagine, Biryani],  
    Beverages = [Tonic, Lassi, _Kombucha, Amasi],  
    pairs_keys_values(Pairs, Chefs, ChefNames),  
    maplist(all_distinct, Table),  
    append(Table, Vs),  
    Vs ins 1..4,
```

Aisha prepares Tagine.

```
Aisha #= Tagine,
```

Emma brings Amasi.

```
Emma #= Amasi,
```

The chef who prepares Frybread brings Tonic.

```
Frybread #= Tonic,
```

Mei brings Lassi.

```
Mei #= Lassi,
```

Winona does not prepare Pad Thai.

```
Winona #\= PadThai,
```

The chef who brings the Lassi did not cook the Biryani

```
Lassi #\= Biryani.
```

dish(Chef, Dish)	garden_party.pl
------------------	-----------------

```
dish(Chef, Dish) :-  
    DishNames = [pad_thai, frybread, tagine, biryani],  
    garden_party(ChefPairs, Dishes, _),  
    label(Dishes),  
    pairs_keys_values(DishPairs, Dishes, DishNames),  
    member((N-Chef), ChefPairs),  
    member((N-Dish), DishPairs),  
    !.
```

beverage(Chef, Beverage)	garden_party.pl
--------------------------	-----------------

```
beverage(Chef, Beverage) :-  
    BeverageNames = [tonic, lassi, kombucha, amasi],  
    garden_party(ChefPairs, _, Beverages),  
    label(Beverages),  
    pairs_keys_values(BeveragePairs, Beverages, BeverageNames),  
    member((N-Chef), ChefPairs),  
    member((N-Beverage), BeveragePairs),  
    !.
```